

Problema del Lot-Sizing: specifica

Il problema che è stato analizzato viene chiamato Lot-Sizing. Si tratta di un problema di *production planning* a una macchina.

I seguenti dati vengono forniti per ogni istanza:

- Un insieme $I = \{0, \dots, m - 1\}$ che rappresenta i tipi di oggetto che possono essere prodotti.
- Un insieme $P = \{0, \dots, p - 1\}$ che rappresenta i periodi temporali (discreti) durante i quali produciamo un oggetto.
- Una matrice $I \times P$ (la chiamo D) tale che $d_t^i = 1$ se un oggetto di tipo i è richiesto al tempo p , $d_t^i = 0$ altrimenti.
- Una matrice $I \times I$ (la chiamo C) tale che c^{ij} rappresenta il costo che va pagato quando la macchina passa dalla produzione di un oggetto di tipo i ad uno di tipo j ($\forall i \ c^{ii} = 0$).
- Un array H di lunghezza m tale che h^i rappresenta il costo dato dal tenere in magazzino un oggetto di tipo i per un periodo (una unità temporale).

Assumendo di dover produrre esattamente un oggetto per ogni periodo, il problema consiste nel determinare quale oggetto vada prodotto dalla macchina in ciascun periodo di tempo. La soluzione quindi può essere rappresentata tramite un array V di lunghezza p , dove ogni V^t rappresenterà il tipo di oggetto da produrre al tempo t .

Si ha un unico constraint di tipo hard che possiamo chiamare **NoBacklog**: ogni oggetto deve essere prodotto non più tardi rispetto a quando è richiesto. In altre parole, per ogni oggetto i e ogni periodo t , il numero di oggetti i prodotti fino al tempo t deve essere maggiore o uguale al numero totale di richieste di i fino al tempo t .

La **funzione obiettivo** da minimizzare è costituita da due elementi:

- **StockingCost**: è la somma dei costi di stoccaggio di tutti gli oggetti. Questi costi vengono generati quando la macchina produce un oggetto al tempo t ma l'oggetto è richiesto solamente "dopo" (ad un tempo $t' > t$).

- **SetupCost:** è la somma dei costi di setup della macchina: vengono generati ogni volta che questa passa dal produrre un tipo di oggetto ad uno diverso. Il primo oggetto prodotto genera un SetupCost nullo.

Riporto qui di seguito una piccola istanza esemplificativa in formato MiniZinc:

```
Periods = 8;
Items = 3;
Demands = [|0, 0, 0, 0, 0, 1, 0, 1
           |0, 0, 1, 1, 0, 0, 1, 1
           |0, 0, 0, 0, 0, 1, 1, 0|];
StockingCosts = [10, 15, 12];
SetupCosts = [|0, 131, 109
              |193, 0, 175
              |101, 136, 0|];
```

con la corrispondente soluzione ottima:

```
[1, 1, 2, 2, 0, 0, 1, 1]
```

di costo 569.

Il modello MiniZinc

Riporto qui il codice diviso in parti, al termine di ogni frammento è presente un commento al codice.

```
include "globals.mzn";

int: Periods;
int: Items;
array[1..Items,1..Periods] of 0..1: Demands;
array[1..Items] of int: StockingCosts;
array[1..Items,1..Items] of int: SetupCosts;
```

Qui vengono inizializzate le strutture che servono a modellare l'input del problema. Inoltre è presente l'istruzione `include` che permetterà poi di utilizzare il vincolo globale *exactly*.

```
array[1..Items,1..Periods] of var 0..1: SolMatrix;           % Solution matrix
array[1..Periods] of var 0..Items-1: Sol;                     % Output array
```

SolMatrix è una matrice $item \times period$ di "0" e "1". s_p^i è uguale a 1 se e solo se nel periodo p produco un item di tipo i . *Sol* è un array di dimensione uguale al numero di periodi ed esprime, per ogni periodo, quale tipo di item viene prodotto.

```
% Schedule the production of exactly one item for each period
constraint
  forall (t in 1..Periods)
    (exactly(1,[SolMatrix[i,t] | i in 1..Items],1));
```

Questo primo insieme di vincoli costringe ad assegnare per ogni periodo di tempo esattamente un tipo di item da produrre. Ho fatto alcune prove esprimendo il vincolo in altri modi (ad esempio, vincolare la somma degli elementi di ogni colonna ad essere uguale a 1) ma i tempi erano sempre peggiori rispetto alla versione con l'*exactly*.

```
% Every item must be produced before the time for which it's requested
constraint
  forall (i in 1..Items, j in 1..Periods)
    (sum (k in 1..j) (Demands[i,k]) <= sum (k in 1..j) (SolMatrix[i,k]));
```

Questi vincoli esprimono **NoBacklog**, cioè la non possibilità di produrre un item in un periodo successivo a quello per il quale è richiesto.

```
% This constraint "links" SolMatrix and Sol
constraint
  forall(t in 1..Periods)
    (sum (i in 1..Items)(i*SolMatrix[i,t]) == Sol[t]+1);
```

Qui *SolMatrix* e *Sol* vengono “collegati” per far corrispondere alla matrice l’array che sarà poi l’effettiva soluzione data in output. Sono state mantenute entrambe le strutture (anche se ridondanti) perchè lavorare solo utilizzando la matrice rallentava di molto il calcolo del valore della funzione obiettivo, mentre utilizzare solo l’array rendeva più lento il lavoro fatto dai primi due insiemi di vincoli presentati precedentemente. Utilizzando entrambe le strutture ho ottenuto i risultati migliori.

```
% Minimize the stocking costs (keeping an already produced item stored)
% and
% the setup costs (switching from producing a certain type of item
% to another)
solve:: int_search([SolMatrix[i,j] | i in 1..Items, j in 1..Periods],
  occurrence, indomain_min, complete)

minimize (
  sum (i in 1..Items)
    (StockingCosts[i] * sum(t in 1..Periods, j in 1..t)
      (SolMatrix[i,j]-Demands[i,j]))
  +
  sum (i in 1..Periods-1) (SetupCosts[Sol[i]+1,Sol[i+1]+1])
);
```

Qui sono presenti le annotazioni per la ricerca e la funzione obiettivo: La ricerca parte assegnando valori alla matrice *SolMatrix* (altre possibilità sono state testate e sono risultate sensibilmente peggiori). L’euristica per la variable selection è *occurrence*, viene scelta la variabile più vincolata.

Anche qui è stato visto sperimentalmente che è la scelta migliore. Per quanto riguarda value choice, *indomain_min* è chiaramente la migliore visto che la mia soluzione corrisponderà ad una matrice nella quale la maggioranza dei valori è 0.

La funzione da minimizzare è la somma degli *stockingcosts* e dei *setupcosts*. Qui si nota anche l'utilità di avere sia *SolMatrix* che *Sol*.

```
output ["Cost: "++show (
  sum (i in 1..Items)
    (StockingCosts[i] * sum(t in 1..Periods, j in 1..t)
      (SolMatrix[i,j]-Demands[i,j]))
  +
  sum (i in 1..Periods-1) (SetupCosts[Sol[i]+1,Sol[i+1]+1])
)++" "]
++
[show(Sol)]
```

L'output è composto dal costo della soluzione trovata e dall'array che esprime la soluzione.